content: use AppleScript version "2.4" -- Yosemite (10.10) or later
  use scripting additions
  use O : script "omnifocus"
  
  tell application "OmniFocus"
  	tell default document
  		repeat with _task in (every inbox task)
  			try
  				set projectName to (name of (every flattened project whose status is active or status is on hold ¬
  					and name of its folder is not ¬
  					"Navigation" and name of its folder is not ¬
  					"Templates" and hidden of its folder is not true))
  				set _project to O's findProject(choose from list projectName with title "Interactive Inbox" with prompt "Assign " & name of _task & " to:" without empty selection allowed)
  				set assigned container of _task to _project
  			on error
  				display notification "No project selected for " & name of _task & "."
  			end try
  		end repeat
  		
  		compact
  		
  		if (count of inbox tasks) > 1 then
  			display notification (((count of inbox tasks) as text) & " tasks remain in the inbox.")
  		else if (count of inbox tasks) > 0 then
  			display notification (((count of inbox tasks) as text) & " task remains in the inbox.")
  		end if
  	end tell
  end tell
  
  